

How do you use the Isources interface?

From OpenDSSWiki

Question

I want to use my own circuit element model instead of the one in OpenDSS. How can I control a current source from MATLAB?

Answer

--Rdugan 14:26, 10 June 2014 (UTC)

Since OpenDSS uses a relatively straightforward **nodal admittance** network model formulation, **active device models** basically inject a current into a bus to represent some behavior. So you can implement your model through an injection source.

You have a couple of main choices:

- Use one of the existing Power Conversion (PC) element such as LOAD or GENERATOR and adjust the **power** injection (kW and kvar properties) as needed for each solution step;
- Connect a current source (ISOURCE object) into the network and control the the **current** magnitude and angle to achieve the desired results.

This post will deal with the latter option. You could always manipulate the ISOURCE objects in the circuit via the text interface. However, we have recently added the **Isources interface** to the COM server at *Version 7.6.3.30* to make it easier to program and more efficient to execute.

Below is an Excel VBA example for controlling an ISOURCE object through the Isources interface. The example is not in MATLAB, but you should get the idea. You will load this into Excel using the VBA editor (alt-F11 will bring it up). Change the path to the IEEE 13-bus test feeder that ships with OpenDSS as needed. Then execute the macro (the Sub). Just click in the routine somewhere then click the run button or press F5.

This example changes the ISOURCE.IS1 magnitude from 0 to 100 A randomly as it changes the phase angle from 0 to 360 degrees. Paste the results in the CSV files that pop up after the solution is completed into Excel worksheets and plot the results. It makes interesting-looking plots.

The example also demonstrates how to use the **Monitors interface**.

```
Option Explicit

! Example VBA Script for testing ISources Interface

Public DSSobj As OpenDSSengine.DSS
Public DSSText As OpenDSSengine.Text
Public DSSCircuit As OpenDSSengine.Circuit
```

```

Public DSSSolution As OpenDSSEngine.Solution
Public DSSControlQueue As OpenDSSEngine.CtrlQueue
Public DSSCktElement As OpenDSSEngine.CktElement
Public DSSPDElement As OpenDSSEngine.PDElements
Public DSSMeters As OpenDSSEngine.Meters
Public DSSBus As OpenDSSEngine.Bus
Public DSSCmath As OpenDSSEngine.CmathLib
Public DSSParser As OpenDSSEngine.Parser
Public DSSIsources As OpenDSSEngine.ISources
Public DSSMonitors As OpenDSSEngine.Monitors

Public Sub StartDSS()

' Create a new instance of the DSS
Set DSSobj = New OpenDSSEngine.DSS

' Start the DSS
If Not DSSobj.Start(0) Then
    MsgBox "DSS Failed to Start"
Else
    ' MsgBox "DSS Started successfully"
    ' Assign a variable to each of the interfaces for easier access
    Set DSSText = DSSobj.Text
    Set DSSCircuit = DSSobj.ActiveCircuit
    Set DSSSolution = DSSCircuit.Solution
    Set DSSControlQueue = DSSCircuit.CtrlQueue
    Set DSSCktElement = DSSCircuit.ActiveCktElement
    Set DSSPDElement = DSSCircuit.PDElements
    Set DSSMeters = DSSCircuit.Meters
    Set DSSBus = DSSCircuit.ActiveBus
    Set DSSCmath = DSSobj.CmathLib
    Set DSSParser = DSSobj.Parser
    Set DSSIsources = DSSCircuit.ISources
    Set DSSMonitors = DSSCircuit.Monitors

    ' Range("DSSVersion").Value = "Version: " + DSSobj.Version
    Beep
End If

End Sub

Public Sub TestISources()

' Example using the Isources interface to control a current source
' Requires version 7.6.3.30 or later

Dim i As Long, iMon As Long

StartDSS

' Compile a DSS circuit model for testing the interface

With DSSText
    .Command = "Compile C:\Users\prdu001\OpenDSS\Distrib\IEEETestCases\13Bus\IEEE13Nodeckt.dss"

    ' Add an Isource (nominal 100 A)
    .Command = "New Isource.IS1 Phases=3 Bus1=675 amps=100 angle=0 frequency=60"

    ' Add some Monitors to capture results
    .Command = "New Monitor.M1 Line.650632 1 Mode=0" 'VI monitor at head of feeder
    .Command = "New Monitor.M2 Isource.IS1 1 mode=1 ppolar=no" 'PQ monitor on the Isource
End With

```

```
' Set random currents up to 100 A as we vary the angle from 0 to 360.

DSSMonitors.Reset

DSSIsources.Name = "IS1" ' make sure Isource.IS1 active

For i = 1 To 360
    DSSSolution.Solve
    DSSSolution.dblHour = i ' a number to put in the time columns
    DSSMonitors.SampleAll
    DSSIsources.Amps = 100# * Rnd
    DSSIsources.AngleDeg = DSSIsources.AngleDeg + 1
Next i

DSSMonitors.SaveAll
iMon = DSSMonitors.First
Do While iMon > 0
    DSSMonitors.Show
    iMon = DSSMonitors.Next
Loop

End Sub
```

Retrieved from "http://localhost/mediawiki/index.php?title=How_do_you_use_the_Isources_interface%3F&oldid=701"

- This page was last modified on 10 June 2014, at 06:42.
- This page has been accessed 18 times.