

OpenDSS STORAGE Element and STORAGECONTROLLER Element

(Version 7.4.1 Build 35 and Later)
(Sourceforge repository version 517 or later)

Revised 5 March-2011

Storage Element

The storage element is essentially a generator that can be dispatched to either produce power (discharge) or consume power (charge) within its power rating and its stored energy capacity. The model was developed from the Generator element model. Thus, it has inherited some of the features such as a built-in energy meter and an interface to user-written DLLs.

A storage element can either act independently or be controlled by a StorageController element.




Figure 1. Basic concept of the Storage Element

The model may be used in a Snapshot power flow mode to simply compute the power flow for a selected *state* of the Storage element flow control (see Figure 1). In that case, you would simply set the state to one of {IDLING | CHARGING | DISCHARGING} and then solve.

Note that there will only be power discharged if the present charge level (kWhStored or %Stored properties) is greater than the specified reserve level (see %Reserve property). The Storage element will only take charge when the kWhStored value is less than kWhRated. You can specify the rate of discharge with the %Discharge value and the rate of charge with the %Charge value.

However, the strength of the model is in time-varying simulation modes. Daily, Yearly, and DutyCycle modes are supported. You would typically use Daily or Yearly modes to look at general energy issues over a period of time with time step sizes of several minutes to one hour. The DutyCycle mode would be used to study the effectiveness of storage to compensate for short term power variations such as might occur in a matter of seconds with a cloud transient affecting solar PV generation.

The storage element can also produce or absorb reactive power (vars) within the kVA rating of the inverter. That is, a StorageController object requests a certain amount of kvar and the storage element provides it if the inverter has any capacity left. The storage element can produce/absorb vars while idling.

Losses are important when evaluating storage schemes. The model allows separate specification of charging and discharging efficiencies. The default values for each direction are 90%, making a nominal round trip efficiency of 81%. Set the charge and discharge efficiency to desired values.

In addition, idling losses may be specified. These represent energy required by the internal controls, heaters, coolers, etc., to maintain proper battery temperatures. This loss is currently specified as a single average value specified in percent of power rating (default = 1%) and is modeled as a constant impedance in shunt with the power system. Of course, your script can change this value on the fly.

In charging or discharging mode, the Storage element is generally modeled as a simple constant ($P+jQ$) model (model=1, the default). A constant Z model (model=2) is also available. The Usermodel support code was ported from the Generator model. This requires a DLL to be written.

Dispatch Modes and Triggers

The user specifies the dispatch mode as one of

{DEFAULT | FOLLOW | EXTERNAL| LOADLEVEL | PRICE }

The **ChargeTrigger** and **DischargeTrigger** values interact differently in each mode. Basically, when a trigger level is surpassed the Storage element is permitted to go into the corresponding operating mode. Whether it actually does depends on several factors.

Default Mode: The triggers follow the specified load shape corresponding to the present solution mode (Daily, Yearly, DutyCycle). For example, if a Yearly solution is being performed, the storage device will discharge at the (fixed) specified discharge rate whenever the Yearly LoadShape object value exceeds the DischargeTrigger value. It will stay in Discharge mode until either the LoadShape value drops below the trigger level or the Storage element depletes its stored energy down to the reserve level. When the LoadShape value drops below the ChargeTrigger value, the Storage element will begin to charge at the specified charging rate. It will continue charging until the storage is built back up to 100% or another discharge cycle is required.

There is a default time for starting the charge cycle (TimeChargeTrigger) that will override the ChargeTrigger criterion if it specified. This is to ensure that Charging will take place even if the load level does not drop below the ChargeTrigger value at any time

during the day. Set TimeChargeTrigger to a negative number to disable a default charging time.

For example, the default charging time is 2 AM. At this time, the Storage element will attempt to charge even if the load has not dropped below the ChargeTrigger value. This is a strategy for ensuring that the Storage element is always fully charged for the next day's peak load.




Figure 2. Illustrating Default Dispatch

The default mode is illustrated in Figure 2. The DischargeTrigger property is set to 0.92 and the ChargeTrigger property to 0.4. When the dispatch loadshape multiplier exceeds 0.92 in this case, the STORAGE element is set to discharge at the presently-defined discharge rate. Likewise, when the dispatch loadshape multiplier drops below 0.40, the charge cycle begins. On days when the loadshape multiplier does not drop below 0.40, the time charge, which defaults to 2 AM, takes and starts the charging at the designated rate.

FOLLOW mode: The kW and kvar output of the STORAGE element follows the active loadshape multipliers until kWh storage is either exhausted or full. The STORAGE element discharges for positive values and charges for negative values. Charging and

discharging are **proportional** to the **kWrated** property. This is illustrated in Figure 3. The Discharge Cycle is set to nominally follow the shape of the daily peak that occurs approximately 5 PM. If you had a 1000 kWh battery with a 250 kW inverter. Over the 4.5 hr period, this would nearly discharge the battery, reaching a peak of 250 kW at 5 PM (hr 17). At 2 AM the next day, the charge cycle would begin at 2 AM ramping up to 250 kW over 0.5 hr and then continuing, if needed, until 6 AM. Obviously, you would have to simulate at least 2 days to see if this would work the way you wanted because the STORAGE unit begins fully charged. Therefore, there would be no charging on the first charge cycle.




Figure 3. Example Daily Charge/Discharge Cycle in Follow Mode

Figure 4 shows a result that might come from the simulation described. The Loadshape in Figure 3 is overlaid on the Power at the terminals of the storage device. A positive value for power denotes discharging in this chart (OpenDSS will show the power as negative because it is coming out of the terminal.)

Note that the 1000 kWh battery discharges to the reserve level of 20% about one hour short of meeting the discharge dispatch goal. The specified charge cycle is a little longer than needed to get the battery back to 100% charge level before charging is prohibited by following the dispatch loadshape.

You could use this daily shape for a yearly simulation. It would simply repeat over and over.

Note that there is no guarantee of charging in Follow mode. The loadshape must be defined negative for a period to force charging. You generally want somewhat more area under the charging curve than the discharge curve to compensate for losses.

If the STORAGE element has insufficient kWh to make it through the discharge cycle, it will switch to idling. Likewise, if the storage element fills before the end of the charge cycle, it will switch to idling. Charging and discharging losses are accounted for during these cycles.




Figure 4. Example Simulation Result

Although not shown, both kW and kvar are dispatched. The kW is dispatched first and then the kvar is dispatched up to the kVA rating of the STORAGE element. As with all Loadshape objects, if Qmult is not specified, the reactive power multiplier defaults to the Mult property value (same as for kW).

The reactive power dispatch during simulations is relative to the value after the most recent setting of the kvar property, or the kW and PF properties. This establishes the base kvar value. The kvar output is determined by multiplying the Qmult value times the base kvar value.

The Loadshape objects used with this mode are usually **Normalized** to 1.0 or another per unit value.

The following script was used to generate the result shown in Figure 4:

```

Clear
New Circuit.TestStorage
~ BasekV=12.47
New Line.Line1 Bus1=Sourcebus LoadBus ! default line
New Loadshape.DailyShape npts=96 mininterval=15 mult=[file=storagetestshape.csv]
New Storage.Battery phases=3 Bus1=loadbus kV=12.47 kW=250 kWrated=250 kWhrated=1000
~ dispmode=follow daily=dailyshape
set voltagebase=[12.47]
calcv
new monitor.PQ storage.battery 1 ppolar=no mode=1
new monitor.Vars storage.battery 1 mode=3
solve
solve mode=daily step=15m number=(2 96 *)
show mon PQ
show mon vars

```

(Note that the dispatch shape in Figure 3 was loaded into the file “storagetestshape.csv”).

LOADLEVEL or PRICE mode: This DischargeTrigger/ChargeTrigger algorithm may also be applied to the global LOADLEVEL or PRICE values instead of the LoadShape objects defined for a particular Storage element.

See the Help for the following Options:

LOADLEVEL is set at a global circuit value with:

| | |
|---|-----------------------------|
| Set LoadMult= <i>perunitvalue</i> | |
| Set DefaultDaily= <i>MyDailyLoadShapeName</i> | (for daily-mode solutions) |
| Set DefaultYearly= <i>MyYearlyLoadshapeName</i> | (for yearly mode solutions) |

PRICE is set to a global circuit value with:

| |
|---|
| Set PriceSignal= <i>value</i> |
| Set PriceCurve= <i>MyPriceCurveName</i> |

As of Ver. 7.4.1 build 35, price curves are defined with a PRICESHAPE object rather than a LoadShape object.

With any of the above three dispatch modes, you can get quite creative defining LoadShape objects in, for example, MS Excel that will allow for the simulation of a wide variety of charge/discharge behaviors. You can use this procedure to simulate the effect of having a central storage controller even if none is defined. By assigning the same loadshapes to a number of Storage elements, you can simulate many aspects of a StorageController handling a fleet of dispersed storage devices.

EXTERNAL mode: The Storage element does not attempt to determine its own state in this mode. Instead, it assumes a StorageController element will provide the required control of the state of the Storage element. This mode is automatically set when a StorageController grabs control of a Storage element.

While you can make up loadshapes to perform offline simulation of load following, etc. a StorageController is required to perform this function in real time simulations or off-line simulations that do not have predefined behaviors.

In addition to these dispatch modes, you may simply control the State of the Storage element explicitly by scripting the simulation through a script file or under program control through the COM interface.

Storage Element Properties

| Property Name | Description |
|-----------------|--|
| (1) phases | Number of Phases, this Storage element. Power is evenly divided among phases. |
| (2) bus1 | Bus to which the Storage element is connected. May include specific node specification. |
| (3) kV | Nominal rated (1.0 per unit) voltage, kV, for Storage element. For 2- and 3-phase Storage elements, specify phase-phase kV. Otherwise, specify actual kV across each branch of the Storage element. If wye (star), specify phase-neutral kV. If delta or phase-phase connected, specify phase-phase kV. |
| (4) kW | Get/set the present kW value. A positive value denotes power coming OUT of the element, which is the opposite of a Load element. A negative value indicates the Storage element is in Charging state. This value is modified internally depending on the dispatch mode. |
| (5) pf | Nominally, the power factor for discharging (acting as a generator). Default is 1.0. Setting this property will also set the kvar property. Enter negative for leading power factor (when kW and kvar have opposite signs.) A positive power factor for a generator signifies that the Storage element produces vars as is typical for a generator. |
| (6) conn | = {wye LN delta LL}. Default is wye. |
| (7) kvar | Get/set the present kvar value. Alternative to specifying the power factor. Side effect: the power factor value is altered to agree based on present value of kvar. |
| (8) kVA | kVA rating of power output. Defaults to rated kW. Used as the base for Dynamics mode and Harmonics mode values. |
| (9) kWrated | kW rating of power output. Side effect: Set KVA property. |
| (10) kWhrated | Rated storage capacity in kWh. Default is 50. |
| (11) kWhstored | Present amount of energy stored, kWh. Default is same as kWh rated. |
| (12) %stored | Present amount of energy stored, % of rated kWh. Default is 100%. |
| (13) %reserve | Percent of rated kWh storage capacity to be held in reserve for normal operation. Default = 20. This is treated as the minimum energy discharge level unless there is an emergency. For emergency operation set this property lower. Cannot be less than zero |
| (14) State | {IDLING CHARGING DISCHARGING} Get/Set present operational state. In DISCHARGING mode, the Storage element acts as a generator and the kW property is positive. The element continues discharging at the scheduled output power level until the storage reaches the reserve value. Then the state reverts to IDLING. In the CHARGING state, the Storage element behaves like a Load and the kW property is negative. The element continues to charge until the max storage kWh is reached and Then switches to IDLING state. In IDLING state, the kW property shows zero. However, the resistive and reactive loss elements remain in the circuit and the power flow report will show power being consumed. |
| (15) %Discharge | Discharge rate (output power) in Percent of rated kW. Default = 100. |

| | |
|--------------------|--|
| (16) %Charge | Charging rate (input power) in Percent of rated kW. Default = 100. |
| (17) %EffCharge | Percent efficiency for CHARGING the storage element. Default = 90. |
| (18) %EffDischarge | Percent efficiency for DISCHARGING the storage element. Default = 90. Idling losses are handled by %IdlingkW property and are in addition to the charging and discharging efficiency losses in the power conversion process inside the unit. |
| (19) %IdlingkW | Percent of rated kW consumed while idling. Default = 1. |
| (20) %Idlingkvar | Percent of kWrated property consumed as reactive power (kvar) while idling. Default = 0. |
| (21) %R | Equivalent percent internal resistance, ohms. Default is 0. Placed in series with internal voltage source for harmonics and dynamics modes. Use a combination of %IdlingkW and %EffCharge and %EffDischarge to account for losses in power flow modes. |
| (22) %X | Equivalent percent internal reactance, ohms. Default is 50%. Placed in series with internal voltage source for harmonics and dynamics modes. (Limits fault current to 2 pu.) Use %Idlekvar and kvar properties to account for any reactive power during power flow solutions. |
| (23) model | Integer code for the model to use for power output variation with voltage. Valid values are: 1:Storage element injects a CONSTANT kW at specified power factor. 2:Storage element is modeled as a CONSTANT ADMITTANCE. 3:Compute load injection from User-written Model. |
| (24) Vminpu | Default = 0.90. Minimum per unit voltage for which the Model is assumed to apply. Below this value, the load model reverts to a constant impedance model. |
| (25) Vmaxpu | Default = 1.10. Maximum per unit voltage for which the Model is assumed to apply. Above this value, the load model reverts to a constant impedance model. |
| (26) yearly | Dispatch shape to use for yearly simulations. Must be previously defined as a Loadshape object. If this is not specified, the Daily dispatch shape, if any, is repeated during Yearly solution modes. In the default dispatch mode, the Storage element uses this loadshape to trigger State changes. |
| (27) daily | Dispatch shape to use for daily simulations. Must be previously defined as a Loadshape object of 24 hrs, typically. In the default dispatch mode, the Storage element uses this loadshape to trigger State changes. |
| (28) duty | Load shape to use for duty cycle dispatch simulations such as for solar ramp rate studies. Must be previously defined as a Loadshape object. Typically would have time intervals of 1-5 seconds. Designate the number of points to solve using the Set Number=xxxx command. If there are fewer points in the actual shape, the shape is assumed to repeat. |
| (29) dispmode | {DEFAULT FOLLOW EXTERNAL LOADLEVEL PRICE } Default = "DEFAULT". Dispatch mode. In DEFAULT mode, Storage element state is triggered by the loadshape curve corresponding to the solution mode. In FOLLOW mode the kW and kvar output of the STORAGE element |

| | |
|-----------------------|--|
| | <p>follows the active loadshape multipliers until storage is either exhausted or full. The element discharges for positive values and charges for negative values. The loadshapes are based on the kW and kvar values in the most recent definition of kW and PF or kW and kvar properties.</p> <p>In EXTERNAL mode, Storage element state is controlled by an external Storage controller. This mode is automatically set if this Storage element is included in the element list of a StorageController element.</p> <p>For the other two dispatch modes, the Storage element state is controlled by either the global default Loadlevel value or the price level.</p> |
| (30) dischargetrigger | <p>Dispatch trigger value for discharging the storage. (except Follow mode)</p> <p>If = 0.0 the Storage element state is changed by the State command or by a StorageController object.</p> <p>If $\neq 0$ the Storage element state is set to DISCHARGING when this trigger level is EXCEEDED by either the specified Loadshape curve value or the price signal or global Loadlevel value, depending on dispatch mode. See State property.</p> |
| (31) Chargetrigger | <p>Dispatch trigger value for charging the storage. (except Follow mode)</p> <p>If = 0.0 the Storage element state is changed by the State command or StorageController object.</p> <p>If $\neq 0$ the Storage element state is set to CHARGING when this trigger level is GREATER than either the specified Loadshape curve value or the price signal or global Loadlevel value, depending on dispatch mode. See State property.</p> |
| (32) TimeChargeTrig | <p>Time of day in fractional hours (0230 = 2.5) at which storage element will automatically go into charge state. Default is 2.0. Enter a negative time value to disable this feature. Does not apply to Follow mode.</p> |
| (33) class | <p>An arbitrary integer number representing the class of Storage element so that Storage values may be segregated by class.</p> |
| (34) UserModel | <p>Name of DLL containing user-written model, which computes the terminal currents for Dynamics studies, overriding the default model. Set to "none" to negate previous setting.</p> |
| (35) UserData | <p>String (in quotes or parentheses) that gets passed to user-written model for defining the data required for that model.</p> |
| (36) debugtrace | <p>{ Yes No } Default is no. Turn this on to capture the progress of the Storage model for each iteration. Creates a separate file for each Storage element named "STORAGE_name.CSV".</p> |
| (37) spectrum | <p>Name of harmonic voltage or current spectrum for this Storage element. Current injection is assumed for inverter. Default value is "default", which is defined when the DSS starts.</p> |
| (38) basefreq | <p>Base Frequency for ratings.</p> |
| (39) enabled | <p>{ Yes No or True False } Indicates whether this element is enabled.</p> |
| (40) like | <p>Make like another object, e.g.:</p> <p>New Storage.S2 like=S1 ...</p> |
| | |

Example Usage

A couple of simple example scripts designed for use with the OpenDSS version of the IEEE 123 bus test feeder are presented. These would work well in the default SnapShot power flow mode. After solving you can execute the visualize command as indicated to readily observe the result.

3-phase 150 kWh, 75 kW unit:

```

! Defining the storage unit at bus 98
New Storage.N98 Bus1=98 kV=4.16 kWRated=75 kWhRated=150 kWhStored=150
~ State=IDLING    !(this is the default)

! Now set it to discharge at 25% rate
Storage.n98.state=Dischar %discharge=25
FormEdit "storage.n98"    ! this will confirm the property changes
Solve
Visualize powers storage.n98    ! now you can see the result

! Now set it to charge at 50% rate
! note that we have to back off the kWhstored so that it will actually
! go into charging mode. Otherwise, it goes to idling mode.
Storage.n98.state=charging kWhstored=100 %charge=50
Solve
Visualize powers storage.n98    ! now you can see the result

```

1-Phase 75 kWh, 25 kW unit

```
New Storage.N98 Phases=1 Bus1=98 kV=2.4 kWRated=25 kWhRated=75  
~ kWhStored=55 State=IDLING
```

```
! Note that this time we defined it with kWhStored < kWhRated  
! so we can put it into either charge or discharge mode
```

```
! set it to discharge at a 25% rate
```

```
Storage.n98.state=Dischar %discharge=25
```

```
Solve
```

```
Visualize powers storage.n98 ! now you can see the result
```

```
! set it to charge at a 50% rate
```

```
Storage.n98.state=charging %charge=50
```

```
Solve
```

```
Visualize powers storage.n98 ! now you can see the result
```

StorageController Element

The StorageController element is designed to control a *fleet* of Storage elements and perform such tasks as dispatching the Storage elements to follow load. It is difficult to perform load following based on local intelligence only at each Storage element location because the local load may not reflect the load at a substation, for example, that needs capacity relief. So the StorageController was created to perform that task. The concept is depicted in Figure 5. The “fleet” may consist of one or more Storage devices.




Figure 5. StorageController Concept

Like other controllers in OpenDSS, you define a StorageController as monitoring a *terminal* of one of the current-carrying devices in the circuit. This is usually a Line or Transformer object, but could be any circuit element including a Generator object representing, for example, a solar PV installation with varying power output.

Another key property is **ElementList**. This is a list of Storage-class elements controlled by this controller. If the list is not defined, **all Storage elements** in the circuit are assumed to be controlled by this controller. You may list the elements as an array of names or you may use the “file=” syntax within the array delimiters:

```
... ElementList = [MyElement1, MyElement2, MyElement3, ... ]
```

```
... ElementList = [File=listfile.txt]
```

Where *listfile.txt* is a file containing the names of the Storage elements one to a line. (Only the name of the element should be given; the class name “Storage” is assumed.)

You will get an error message if more than one StorageController attempts to control the same Storage device.

Each Storage element may have a different weight, if you want the StorageController to dispatch some elements at a higher rate than others. The weights default to 1.0, but may be defined by an array. While Storage elements are dispatched by percent of rating and the use of weights is likely to be rare, this feature is available.

Operating Modes

There are 5 discharge modes and 2 charging modes.

Discharge Modes

The discharge modes (**ModeDischarge** property) are

{Peakshave | Follow | Support | Loadshape | Time}

The modes are defined as follows:

- **Peakshave:** PeakShave is the default discharge mode. The control attempts to discharge storage to keep power in the monitored element within a bandwidth (specified by the %kWBand property) of, or below, the kW value specified by the kWTarget property. The storage fleet turns itself off when it runs out of stored energy. The basic concept is to follow the load by keeping the load in the monitored terminal at or *below* the kWtarget value. The monitored branch would typically be a line or transformer that might be considered overloaded at the kWtarget value. (Note that it is possible to change kWtarget on the fly for real time control simulation.)
- **Follow:** Load following mode. The control is triggered by time (see TimeDischarge Trigger Property) and resets the kWTarget value to the present monitored element power. It then attempts to discharge storage to keep power in the monitored element no greater than the new kWTarget. Once the fleet is set to discharge, the behavior is essentially the same as Peakshave mode with the exception that there is a different kWTarget value.
- **Support:** DR Support mode. This essentially the opposite of Peakshave mode. Storage is dispatched to keep the power in the monitored terminal at or *above* the kWTarget value. The typical application would be to support renewable generation when power output falls off to events like cloud transients, etc.
- **LoadShape:** In this mode both discharging and charging precisely follow a *per unit* LoadShape curve. When the value is positive all units are discharged at the rate of the per unit value of the curve. When the value is zero, all units are set to idling. When the value is negative, all units are set to charging at the rate of the per unit value of the curve.
- **Time:** In Time mode all storage elements are set to discharge when in the course of simulation the time of day passes the specified hour of day by the

TimeDisChargeTrigger property. (Hour is a decimal value, e.g., 10.5 = 1030.) The storage elements shut themselves off when their storage declines to the reserve value. The discharging rate is set by the %RateDisCharge property.

Charge Modes

Two of the Discharge modes, LoadShape and Time, also apply to charging (ModeCharge). In the charging context, these modes are defined as follow:

- **Loadshape:** See explanation for discharging above; follows the same logic.
- **Time:** In Time mode all storage elements are set to charge when in the course of simulation the time of day passes the specified hour of day by the TimeChargeTrigger. (Hour is a decimal value, e.g., 2.5 = 0230.) The storage elements switch to idling when the storage is completely charged or the mode is changed by the controller. This is the default mode for charging. The charging rate is set by the %RateCharge property.

Deadband Control

Peakshave, Follow, and Support modes both employ a deadband controller concept. A target value is set (kWTarget or PFTarget) and a band around the target is specified. In load following mode, when the kW in the monitored branch exceeds the kWTarget value plus half the bandwidth, storage is dispatched to bring the power back to the center of the band. The dispatch follows the load until the storage dispatch kW wants to go negative, at which point the Storage elements are set to idling.

The band on the kW is set using the %kWBand property, which defaults to 2%.

The PFTarget value (default is 0.96) is used to compute the dispatch of the reactive power. The kvar are dispatched to bring the monitored branch back to the desired power factor band. Default for the PFBand is 0.04. This should not be set extremely tight. To prevent kvar dispatch, set PFBand wider than the actual deviation from the PFTarget value.

Specify PFTarget as positive for lagging load or DR producing vars. Negative values imply that the watts and vars are flowing in opposite directions.

Properties

| Property | Description |
|-------------------|---|
| (1) Element | Full object name of the circuit element, typically a line or transformer, which the control is monitoring. There is no default; must be specified. |
| (2) Terminal | Number of the terminal of the circuit element to which the StorageController control is connected. 1 or 2, typically. Default is 1. Make sure you have the direction on the power matching the sign of kWLimit. |
| (3) kWTarget | kW target for Discharging. The storage element fleet is dispatched to try to hold the power in band at least until the storage is depleted. |
| (4) %kWBand | Bandwidth (% of Target kW) of the dead band around the kW target value. Default is 2% (+/-1%).No dispatch changes are attempted If the power in the monitored terminal stays within this band. |
| (5) PFTarget | Power Factor target for dispatching the reactive power. Default is 0.96. The reactive power of the storage element fleet is dispatched to try to hold the power factor in band. It is assumed that the storage element inverter can produce kvar up to its kVA limit regardless of storage level. |
| (6) PFBand | Bandwidth of the Target power factor of the monitored element. of the dead band around the kvar target value. Default is 0.04 (+/- 0.02).No dispatch changes of the kvar are attempted If the power factor of the monitored terminal stays within this band. |
| (7) ElementList | Array list of Storage elements to be controlled. If not specified, all storage elements in the circuit are assumed dispatched by this controller. |
| (8) Weights | Array of proportional weights corresponding to each storage element in the ElementList. The needed kW or kvar to get back to center band is dispatched to each storage element according to these weights. Default is to set all weights to 1.0. |
| (9) ModeDischarge | {Peakshave Follow Support Loadshape Time} Mode of operation for the DISCHARGE FUNCTION of this controller. In PeakShave mode (Default), the control attempts to discharge storage to keep power in the monitored element below the kWTarget. In Follow mode, the control is triggered by time and resets the kWTarget value to the present monitored element power. It then attempts to discharge storage to keep power in the monitored element below the new kWTarget. See TimeDischargeTrigger. In Support mode, the control operates oppositely of PeakShave mode: storage is discharged to keep kW power output up near the target. In Loadshape mode, both charging and discharging precisely follow the per unit loadshape. Storage is discharged when the loadshape value is positive. In Time mode, the storage discharge is turned on at the specified %RatekW and %Ratekvar at the specified discharge trigger time in fractional hours. |
| (10) ModeCharge | {Loadshape Time} Mode of operation for the CHARGE |

| | |
|------------------------------|--|
| | FUNCTION of this controller. In Loadshape mode, both charging and discharging precisely follows the per unit loadshape. Storage is charged when the loadshape value is negative. In Time mode, the storage charging FUNCTION is triggered at the specified %RateCharge at the specified sharge trigger time in fractional hours. |
| (11) TimeDisChargeTrigger | Default time of day (hr) for initiating Discharging of the fleet. During Follow or Support modes discharging at a fixed time will occur each day at this hour. Set this to a negative value to ignore. Default is -1 (ignored). |
| (12) TimeChargeTrigger | Default time of day (hr) for initiating charging in Time control mode. Set this to a negative value to ignore. Default is 2.0. (0200). When this value is >0 the storage fleet is set to charging at this time regardless of other control criteria to make sure storage is topped off for the next discharge cycle. |
| (13) %RatekW | Sets the kW discharge rate in % of rated capacity for each element of the fleet. Applies to TIME control mode or anytime discharging is triggered by time. |
| (14) %Ratekvar | Sets the kvar discharge rate in % of rated capacity for each element of the fleet. Applies to TIME control mode or anytime discharging is triggered by time. |
| (15) %RateCharge | Sets the kW charging rate in % of rated capacity for each element of the fleet. Applies to TIME control mode and anytime charging mode is entered due to a time trigger. |
| (16) %Reserve | Use this property to change the % reserve for each storage element under control of this controller. This might be used, for example, to allow deeper discharges of storage or in case of emergency operation to use the remainder of the storage element. |
| (17) kWhTotal | (Read only). Total rated kWh energy storage capacity of storage elements controlled by this controller. |
| (18) kWTotal | (Read only). Total rated kW power capacity of storage elements controlled by this controller. |
| (19) kWhActual | (Read only). Actual kWh output of all controlled storage elements. |
| (20) kWActual | (Read only). Actual kW output of all controlled storage elements. |
| (21) kWneed | (Read only). KW needed to meet target. |
| (22) %Participation | Participation factor, %. Default = 100. |
| (23) Yearly | Dispatch loadshape object, If any, for Yearly solution Mode. |

| | |
|---------------|---|
| (24) Daily | Dispatch loadshape object, If any, for Daily solution mode. |
| (25) Duty | Dispatch loadshape object, If any, for Dutycycle solution mode. |
| (26) basefreq | Base Frequency for ratings. |
| (27) enabled | { Yes No or True False } Indicates whether this element is enabled. |
| (28) like | Make like another object, e.g.: New Capacitor.C2 like=c1 ... |

Example Usage

Example 1

```
! discharge = peakshave at 7 MW ! charge=time @30%
New StorageController.CESmain element=line.568_4921721 terminal=1
~ kWTarget=7000 PFTarget=0.98 %ratecharge=30
~ eventlog=y
```

The StorageController is named “CESmain” and is monitoring terminal 1 of a Line element named “568_4921721”.

Since no ModeDischarge property is specified, the default of “Peakshave” is assumed. The controller will attempt to discharge the Storage fleet to maintain the power in the monitored line, which is the first line section in a feeder, to 7MW (7000 kW). The dead band on this value is the default value of 2%, or 140 kW (+/- 70 kW).

The charging rate is set to 30% and the charge cycle will be dispatched at the default value of 0200.

Actions taken by this controller will be recorded on the Event Log (which may be seen by the Show Events command).

Example 2

```
! discharge = load follow at Noon ! charge=time
New StorageController.CESmain element=line.568_4921721 terminal=1
~ kWTarget=7000 PFTarget=0.98
~ %ratecharge=30
~ eventlog=y
~ timedischargetrigger=12
~ modedischarge=Follow
```

This is basically the same as Example 1 except that we’ve changed the discharge mode to “Follow” and set the discharge time trigger to 12 noon. The controller will sample the load at 12 noon on each day and reset kWTarget to that value. It then will attempt to shave the peak at that value.

In this mode, one gets a little more value out of the storage each day by discharging it near the peak. In simple Peakshave mode, the storage may only be dispatched a few days per year.

Example 3

```
new Storage.battery phases=3 bus1=loadbus kv=12.47
~ kwrated=250 kwhrated=1000
new StorageController.battery element=line.Line1
~ %Reserve=20.0
~ ModeCharge=Loadshape
~ ModeDischarge=Loadshape
~ Yearly=ls_bat_cntrl
~ EventLog=Yes
```

The controller sets the fleet Reserve value to 20%. (In this case, the fleet consists of a single 250kW, 1 MWh battery.)

Both the charge and discharge modes are set to “Loadshape”. That is, they will follow a specified loadshape (Figure 6). The battery will charge for the hours when the loadshape is negative at the rate defined by the curve and attempt to discharge when the value is positive. (It should be obvious that the battery will be discharged to the reserve level well before the end of the loadshape.)




Figure 6. Battery control Loadshape

This Loadshape was computed by another program based on the amount of power available for charging and the necessary power to shave a load peak..

Note that it is unnecessary to set to charge mode to Loadshape. This will be set when the discharge mode is set to Loadshape.